Maple-assisted proof of formula for A202902

Robert Israel

9 May 2018

There are $2^8 = 256$ configurations for a 2×4 sub-array, but the 32 where the top left entry is 1 and its two neighbours are 0 are not allowed. This leaves 224 allowed configurations. Consider the 224×224 transition matrix T such that $T_{ij} = 1$ if the bottom two rows of a 3×4 sub-array could be in configuration i while the top two rows are in configuration j (i.e. the middle row is compatible with both i and j, and every 1 in the middle row has a NW, E or S neighbour that is 1), and 0 otherwise. The following Maple code computes it. Configurations are encoded as 8-element lists in the order

$$\begin{bmatrix} 1 & 2 & 3 & 4 \\ 5 & 6 & 7 & 8 \end{bmatrix}$$

```
> Configs:= select(t -> t[1]=0 or t[2]=1 or t[5]=1, [seq(convert
  (2^8+i,base,2)[1..8],i=0..2^8-1)]):
> Compatible:= proc(i,j)
    if Configs[i][1..4] <> Configs[j][5..8] then return 0 fi;
    if Configs[i][2] = 1 and Configs[j][1]+Configs[i][3]+Configs[i]
  [6] = 0 then return 0 fi;
    if Configs[i][3] = 1 and Configs[j][2]+Configs[i][4]+Configs[i]
  [7] = 0 then return 0 fi;
    if Configs[i][4] = 1 and Configs[j][3]+Configs[i][8] = 0 then
  return 0 fi;
    1
  end proc:
> T:= Matrix(224,224,Compatible):
```

Thus $a(n) = u\, T^n\, v$ where u and v are row and column vectors respectively with $u_i = 1$ for i corresponding to configurations with bottom row $(0, 0, 0)$, 0 otherwise, and $v_i = 1$ for i corresponding to configurations with top row $(0, 0, 0)$, 0 otherwise. The following Maple code produces these vectors.

```
> u:= Vector[row](224, i -> `if`(Configs[i][5..8] = [0,0,0,0],1,0))
  :
  v:= Vector(224, j -> `if`(Configs[j][1..4] = [0,0,0,0],1,0)):
```

To check, here are the first few entries of our sequence.

```
> TV[0]:= v:
  for n from 1 to 14 do TV[n]:= T . TV[n-1] od:
> A:= [seq(u . TV[n],n=1..14)];
```

$$A := [1, 40, 494, 4892, 51068, 538672, 5654616, 59369072, 623600944, 6549786560, 68792261728, 722531010240, 7588808329152, 79705877679872] \quad \textbf{(1)}$$

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$$P := t \mapsto t^{15} - 16\,t^{14} + 76\,t^{13} - 272\,t^{12} + 1060\,t^{11} - 2704\,t^{10} + 5184\,t^9 - 9920\,t^8 + 11904\,t^7 - 9472\,t^6 + 7168\,t^5 - 4096\,t^4 + 1024\,t^3 \quad \textbf{(2)}$$

This turns out to have degree 15, but with the 3 lowest coefficients 0. Thus for $k \geq 0$ we will have

$$0 = u\,P(T)\,T^k\,v = \sum_{i=0}^{15} p_i\,a(i+k)$$

$$= 1024\,a(k+3) - 4096\,a(k+4) + 7168\,a(k+5) - 9472\,a(k+6) + 11904\,a(k+7) - 9920\,a(k+8) + 5184\,a(k+9) - 2704\,a(k+10) + 1060\,a(k+11) - 272\,a(k+12) + 76\,a(k+13) - 16\,a(k+14) + a(k+15)$$

where p_i is the coefficient of t^i in $P(t)$. That is, the sequence certainly satisfies the conjectured recurrence

$$a(n) = 16\,a(n-1) - 76\,a(n-2) + 272\,a(n-3) - 1060\,a(n-4) + 2704\,a(n-5) - 5184\,a(n-6) + 9920\,a(n-7) - 11904\,a(n-8) + 9472\,a(n-9) - 7168\,a(n-10) + 4096\,a(n-11) - 1024\,a(n-12)$$

for $n \geq 15$. All we need to do now is to verify that in fact it satisfies the recurrence for $n = 13$ and $n = 14$ as well.

```
> seq(A[n]-(16*A[n-1] -76*A[n-2] +272*A[n-3] -1060*A[n-4] +2704*A
  [n-5] -5184*A[n-6] +9920*A[n-7] -11904*A[n-8] +9472*A[n-9] -7168*
  A[n-10] +4096*A[n-11] -1024*A[n-12]),n=13..14);
```

0, 0 **(3)**